
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.1.1.1.1 which is part of the NR5GC test suite in the ‘iwd-TTCN3-B2018-12_D19wk12_5G’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
8.1.1.1.1
Test Group:
\NR5GC\8_1
ATS Version:
iwd-TTCN3-B2018-12_D19wk12_5G
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
HiSilicon Balong 5000
Verification Status:
PASS


4. Corrections required for NR5GC test case 8.1.1.1.1
In addition to the change given below, the changes described in CR R5s190150 are required for this test case.

Change 4.1 

	Function name
	f_GetTestcaseAttrib_EarlyContentionResolution()

	Reason for change
	For this test case, RRCSetup for msg4 should not be preconfigured at the SS, so that RRCResume can be sent at step 13

	Summary of change
	Add this test case to the list of test cases requiring early contention resolution

	TTCN module
	TestcaseProperties

	MCC160 Comment
	


Before Change
function f_GetTestcaseAttrib_EarlyContentionResolution(charstring p_Testcase) return boolean

  {

    select (p_Testcase) {

      case ("TC_7_1_1_1_3_NR5GC") { return true; }

      case ("TC_8_1_1_2_1_NR5GC") { return true; } 

    }

    return false;

  }
After Change
function f_GetTestcaseAttrib_EarlyContentionResolution(charstring p_Testcase) return boolean

  {

    select (p_Testcase) {

      case ("TC_7_1_1_1_3_NR5GC") { return true; }

      case ("TC_8_1_1_2_1_NR5GC") { return true; }

      case ("TC_8_1_1_1_1_NR5GC") { return true; }
    }

    return false;

  }
Change 4.2
	Function name
	Altstep a_AspFromSystemAdaptorHandler_NRNG

	Reason for change
	Current NAS emulator implementation is assuming RRCResumeComplete always contains dedicatedNASMessage, but this is optional.

	Summary of change
	Add check to allow for “dedicatedNasMessage” omitted in RRCResumeComplete

	TTCN module
	NAsEmu5G_Main_NRNG

	MCC160 Comment
	


Before Change
altstep a_AspFromSystemAdaptorHandler_NRNG(inout NG_NasSecurity_Type p_NG_NasSecurityByRef) runs on NASEMU_NR_BASE_PTC

  {

    ……………………
        if (ischosen(v_RrcPdu.message_.c1)) {

          if (ischosen(v_RrcPdu.message_.c1.rrcSetupComplete)) {

            // RRC SETUP COMPLETE

            if (ischosen(v_RrcPdu.message_.c1.rrcSetupComplete.criticalExtensions.rrcSetupComplete)) {

              v_NAS_DedicatedMsg[0] := v_RrcPdu.message_.c1.rrcSetupComplete.criticalExtensions.rrcSetupComplete.dedicatedNAS_Message;

              v_NoOfNasPdus := 1;

            }

          } else if (ischosen(v_RrcPdu.message_.c1.ulInformationTransfer)) {

            // UL INFORMATION TRANSFER

            if (ischosen(v_RrcPdu.message_.c1.ulInformationTransfer.criticalExtensions.ulInformationTransfer)) {

              v_ULInformationTransfer := v_RrcPdu.message_.c1.ulInformationTransfer.criticalExtensions.ulInformationTransfer;

              v_NAS_DedicatedMsg[0] := v_ULInformationTransfer.dedicatedNAS_Message;

              v_RrcPduPresent := false;

              v_NoOfNasPdus := 1;

            }

          } else if (ischosen(v_RrcPdu.message_.c1.rrcResumeComplete)) {

            // RRC RESUME COMPLETE

            if (ischosen(v_RrcPdu.message_.c1.rrcResumeComplete.criticalExtensions.rrcResumeComplete)) {

              v_NAS_DedicatedMsg[0] := v_RrcPdu.message_.c1.rrcResumeComplete.criticalExtensions.rrcResumeComplete.dedicatedNAS_Message;

              v_NoOfNasPdus := 1;

            }

          } else {

            // OTHER RRC PDU

          }

        } else { // c1 only option at the moment

        }

………………..
  }
After Change
altstep a_AspFromSystemAdaptorHandler_NRNG(inout NG_NasSecurity_Type p_NG_NasSecurityByRef) runs on NASEMU_NR_BASE_PTC

  {

    ……………………
        if (ischosen(v_RrcPdu.message_.c1)) {

          if (ischosen(v_RrcPdu.message_.c1.rrcSetupComplete)) {

            // RRC SETUP COMPLETE

            if (ischosen(v_RrcPdu.message_.c1.rrcSetupComplete.criticalExtensions.rrcSetupComplete)) {

              v_NAS_DedicatedMsg[0] := v_RrcPdu.message_.c1.rrcSetupComplete.criticalExtensions.rrcSetupComplete.dedicatedNAS_Message;

              v_NoOfNasPdus := 1;

            }

          } else if (ischosen(v_RrcPdu.message_.c1.ulInformationTransfer)) {

            // UL INFORMATION TRANSFER

            if (ischosen(v_RrcPdu.message_.c1.ulInformationTransfer.criticalExtensions.ulInformationTransfer)) {

              v_ULInformationTransfer := v_RrcPdu.message_.c1.ulInformationTransfer.criticalExtensions.ulInformationTransfer;

              v_NAS_DedicatedMsg[0] := v_ULInformationTransfer.dedicatedNAS_Message;

              v_RrcPduPresent := false;

              v_NoOfNasPdus := 1;

            }

          } else if (ischosen(v_RrcPdu.message_.c1.rrcResumeComplete)) {

            // RRC RESUME COMPLETE

            if (ischosen(v_RrcPdu.message_.c1.rrcResumeComplete.criticalExtensions.rrcResumeComplete)) {
                     if (ispresent(v_RrcPdu.message_.c1.rrcResumeComplete.criticalExtensions.rrcResumeComplete.dedicatedNAS_Message)) {
                   v_NAS_DedicatedMsg[0] := v_RrcPdu.message_.c1.rrcResumeComplete.criticalExtensions.rrcResumeComplete.dedicatedNAS_Message;

                   v_NoOfNasPdus := 1;
              }
            }

          } else {

            // OTHER RRC PDU

          }

        } else { // c1 only option at the moment

        }

………………..
  }
Change 4.2
	Function name
	f_NR_RRC_InactiveDef()

	Reason for change
	According to 33.501 6.8.2.1.2 State transition from RRC CONNECTED to RRC INACTIVE:
The UE shall delete the current AS keys KRRCenc, KUPenc (if available), and KUPint (if available), but keep the current AS key KRRCint key

According to 33.501 State transition from RRC_INACTIVE to RRC_CONNECTED to a new gNB/ng-eNB
For protection of all RRC messages except RRCReject message following the sent RRCResumeRequest message, the UE shall derive a KNG-RAN* using the target PCI, target ARFCN-DL/EARFCN-DL and the KgNB/NH based on either a horizontal key derivation or a vertical key derivation as defined in clause 6.9.2.1.1 and Annex A.11/Annex A.12. The UE shall further derive KRRCint, KRRCenc, KUPenc (optionally), and KUPint (optionally) from the newly derived KNG-RAN*.
Following release of AS security in f_NR_RRCRelease_Local() on transition to RRC INACTIVE, AS security should be reestablished with newly derived keys to allow for the UE transition back to RRC CONNECTED.


	Summary of change
	Derive new keys and re-activate AS security following RRC release.

	TTCN module
	NR5GC_RRCSteps

	MCC160 Comment
	


Before Change
function f_NR_RRC_InactiveDef(NR_CellId_Type p_CellId ) runs on NR_BASE_PTC

  {

    //NR FFS can be extended with more parameters...

    var CellIdentity v_CellIdentity := f_NR_CellInfo_GetCellIdentity(p_CellId);

    f_NR_RRCRelease_Common (p_CellId, cs_38508_RRCRelease ( tsc_NR_RRC_TI_Def, cs_38508_SuspendConfig(cs_RAN_NotificationAreaInfo_1Cell ( v_CellIdentity))));

  }
After Change
function f_NR_RRC_InactiveDef(NR_CellId_Type p_CellId ) runs on NR_BASE_PTC

  {

    //NR FFS can be extended with more parameters...

    var CellIdentity v_CellIdentity := f_NR_CellInfo_GetCellIdentity(p_CellId);
    var ARFCN_ValueNR arcfn;

    var NR_SecurityParams_Type v_SecurityParams;
    f_NR_RRCRelease_Common (p_CellId, cs_38508_RRCRelease ( tsc_NR_RRC_TI_Def, cs_38508_SuspendConfig(cs_RAN_NotificationAreaInfo_1Cell ( v_CellIdentity))));
    arcfn := f_NR_CellInfo_GetFrequencySSB(nr_Cell1);    

    v_SecurityParams := f_NR_Security_Get();    

    v_SecurityParams := f_NR_InitAS_KeyChaining_KeyRefresh(0 ,arcfn, v_SecurityParams, 0);        

    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(nr_Cell1, false);

    f_NR_SS_RRC_EnableIntProt_CiphULandDL(nr_Cell1,

                                          v_SecurityParams.AS_Integrity,

                                          v_SecurityParams.AS_Ciphering); 

    f_NR_Security_Set(v_SecurityParams);        
  }
Change 4.3
	Function name
	New function f_NR_InitAS_KeyChaining_KeyRefresh()

	Reason for change
	According 33.501 6.8.2.1.2 State transition from RRC CONNECTED to RRC INACTIVE:

The UE shall delete the current AS keys KRRCenc, KUPenc (if available), and KUPint (if available), but keep the current AS key KRRCint key

Accorfing 33.501 State transition from RRC_INACTIVE to RRC_CONNECTED to a new gNB/ng-eNB
For protection of all RRC messages except RRCReject message following the sent RRCResumeRequest message, the UE shall derive a KNG-RAN* using the target PCI, target ARFCN-DL/EARFCN-DL and the KgNB/NH based on either a horizontal key derivation or a vertical key derivation as defined in clause 6.9.2.1.1 and Annex A.11/Annex A.12. The UE shall further derive KRRCint, KRRCenc, KUPenc (optionally), and KUPint (optionally) from the newly derived KNG-RAN*.
Following release of AS security in f_NR_RRCRelease_Local() on transition to RRC INACTIVE, AS security should be reestablished with newly derived keys to allow for the UE transition back to RRC CONNECTED.



	Summary of change
	New function f_NR_InitAS_KeyChaining_KeyRefresh() (based on f_EUTRA_InitAS_KeyChaining_KeyRefresh()) for horizontal or vertical key derivation.

	TTCN module
	

	MCC160 Comment
	


New function
function f_NR_InitAS_KeyChaining_KeyRefresh(PhysCellId p_TargetPhysicalCellIdentity,

                                              ARFCN_ValueNR p_ARFCN_DL,  

                                              NR_SecurityParams_Type p_Auth_Params,

                                              NextHopChainingCount p_NCC) return NR_SecurityParams_Type

  { 

    var NR_SecurityParams_Type v_Auth_Params := p_Auth_Params;

    // initialise to received as parameter

    if (v_Auth_Params.NCC == p_NCC) {  // Refresh of KeNB; Horizontal key derivation

      v_Auth_Params.KGNB := f_NG_Authentication_A11(v_Auth_Params.KDF,

                                                    v_Auth_Params.KGNB,

                                                    v_Auth_Params.NH,

                                                    p_TargetPhysicalCellIdentity, p_ARFCN_DL);          // @sic R5s100138, R5s160711 sic@                                          

    }

    else {  //Chaining of KgNB; Vertical Key derivation

      while (v_Auth_Params.NCC != p_NCC)  {

        if (v_Auth_Params.NCC != 0) { // NH for NCC =1 is already derived in initial AS Key derivation

          v_Auth_Params.NH := f_NR_Authentication_A10(v_Auth_Params.KDF,

                                                      v_Auth_Params.KAMF,

                                                      v_Auth_Params.NH); //@sic R5s100496, R5s160711 sic@

          // Derive NH and store in v_Auth_Params.KENB

        }

        v_Auth_Params.NCC := (v_Auth_Params.NCC + 1) mod 4;

      }

      v_Auth_Params.KGNB := f_NG_Authentication_A11(v_Auth_Params.KDF,

                                                    v_Auth_Params.KGNB,

                                                    v_Auth_Params.NH,

                                                    p_TargetPhysicalCellIdentity,

                                                    p_ARFCN_DL, true);    // @sic R5s100138 R5s100496 R5s160711 sic@

    }

    // Derive KRRC enc key

    v_Auth_Params.AS_Ciphering.KRRCenc := f_NG_Authentication_A8(tsc_RRC_Enc_Alg,

                                                                 f_NR_ConvertCiphAlg2Bitstring(v_Auth_Params.AS_Ciphering.Algorithm),

                                                                 v_Auth_Params.KGNB,

                                                                 v_Auth_Params.KDF);

    // Derive KRRC int key

    v_Auth_Params.AS_Integrity.KRRCint := f_NG_Authentication_A8(tsc_RRC_Int_Alg,

                                                                 f_NR_ConvertIntAlg2Bitstring(v_Auth_Params.AS_Integrity.Algorithm),

                                                                 v_Auth_Params.KGNB,

                                                                 v_Auth_Params.KDF);

    // Derive UP enc key;

    v_Auth_Params.AS_Ciphering.KUPenc := f_NG_Authentication_A8(tsc_UP_Enc_Alg,

                                                                f_NR_ConvertCiphAlg2Bitstring(v_Auth_Params.AS_Ciphering.Algorithm),

                                                                v_Auth_Params.KGNB,

                                                                v_Auth_Params.KDF);

    // Derive UP int key;

    v_Auth_Params.AS_Integrity.KUPint := f_NG_Authentication_A8(tsc_N_UP_Int_Alg,

                                                                f_NR_ConvertIntAlg2Bitstring(v_Auth_Params.AS_Integrity.Algorithm),

                                                                v_Auth_Params.KGNB,

                                                                v_Auth_Params.KDF);

    return v_Auth_Params;

  };
Change 4.3
	Function name
	

	Reason for change
	According 33.501 6.8.2.1.2 State transition from RRC CONNECTED to RRC INACTIVE:

The UE shall delete the current AS keys KRRCenc, KUPenc (if available), and KUPint (if available), but keep the current AS key KRRCint key

Accorfing 33.501 State transition from RRC_INACTIVE to RRC_CONNECTED to a new gNB/ng-eNB
For protection of all RRC messages except RRCReject message following the sent RRCResumeRequest message, the UE shall derive a KNG-RAN* using the target PCI, target ARFCN-DL/EARFCN-DL and the KgNB/NH based on either a horizontal key derivation or a vertical key derivation as defined in clause 6.9.2.1.1 and Annex A.11/Annex A.12. The UE shall further derive KRRCint, KRRCenc, KUPenc (optionally), and KUPint (optionally) from the newly derived KNG-RAN*.
Following release of AS security in f_NR_RRCRelease_Local() on transition to RRC INACTIVE, AS security should be reestablished with newly derived keys to allow for the UE transition back to RRC CONNECTED.



	Summary of change
	New function for KgNB derivation function as per Annex A.11 of 33.501

	TTCN module
	

	MCC160 Comment
	


New function
function f_NG_Authentication_A11(KDF_Type p_KDF,

                                   B256_Type p_KGNB,

                                   B256_Type p_NH,

                                   PhysCellId p_PhysCellId,

                                   ARFCN_ValueNR p_ARFCN_DL,

                                   boolean p_FromNH := false) return B256_Type // False for Deriving Kgnb* from Kgnb, else from NH     

  { 

    const octetstring const_S70_FC :='70'O; //@sic R5s100178 sic@

    var octetstring v_S;

    var B256_Type v_SourceKey;

    // Generation of String

    v_S := const_S70_FC;

    //FC = 0x70

    v_S :=   ( v_S & int2oct ( p_PhysCellId , 2 ) );

    //P0 = PCI (target physical cell id)

    v_S :=   ( v_S & '0002'O );

    //L0 = length of PCI (i.e. 0x00 0x02)

    v_S :=   ( v_S & int2oct ( p_ARFCN_DL , 3 ) );

    //P1 = ARFCN-DL (target physical cell downlink frequency)

    v_S :=   ( v_S & '0003'O );

    //L1 length of ARFCN-DL 

    if ( p_FromNH == false) // Kgnb* derived from Kgnb

      {

        v_SourceKey := p_KGNB; 

      }

    else // Kgnb* derived from Fresh Derived NH

      {

        v_SourceKey := p_NH; 

      };

    return fx_KeyDerivationFunction ( p_KDF, v_SourceKey, v_S );

  };
Change 4.3
	Function name
	f_NR_RRC_CipherActTime_GetInitial()

	Reason for change
	According 33.501 6.8.2.1.2 State transition from RRC CONNECTED to RRC INACTIVE:

The UE shall delete the current AS keys KRRCenc, KUPenc (if available), and KUPint (if available), but keep the current AS key KRRCint key

Accorfing 33.501 State transition from RRC_INACTIVE to RRC_CONNECTED to a new gNB/ng-eNB
For protection of all RRC messages except RRCReject message following the sent RRCResumeRequest message, the UE shall derive a KNG-RAN* using the target PCI, target ARFCN-DL/EARFCN-DL and the KgNB/NH based on either a horizontal key derivation or a vertical key derivation as defined in clause 6.9.2.1.1 and Annex A.11/Annex A.12. The UE shall further derive KRRCint, KRRCenc, KUPenc (optionally), and KUPint (optionally) from the newly derived KNG-RAN*.
Following release of AS security in f_NR_RRCRelease_Local() on transition to RRC INACTIVE, AS security should be reestablished with newly derived keys to allow for the UE transition back to RRC CONNECTED.



	Summary of change
	Modified function f_NR_RRC_CipherActTime_GetInitial() to add an extra parameter to avoid the incrementing of UL and DL SQN for SRB1 (currently done to allow for exchange of SecurityModeCommand / SecurityModeComplete), in order to allow immediate activation of ciphering. 

Note : Since this is used immediately following re-establishment of SRB1 and SRB2 in f_NR_RRCRelease_Local() an alternative implementation is to use a new template for NR_SecurityActTimeList_Type which hardcodes UL and DL SQN as 0 for SRB1 and SRB2. Another alternative is to make “ActTimeList” optional in NR_AS_CipheringInfo_Type with “omit” having the meaning that ciphering is activated immediately (as is already the case in NR_AS_IntegrityInfo_Type).

	TTCN module
	

	MCC160 Comment
	


Before Change
function f_NR_RRC_CipherActTime_GetInitial(NR_CellId_Type p_CellId) runs on NR_BASE_PTC return NR_SecurityActTimeList_Type

  {

    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList;

    var NR_PdcpCountInfo_Type v_PdcpCountInfo;

    var NR_PDCP_ActTime_Type v_PDCP_ActTimeUL;

    var NR_PDCP_ActTime_Type v_PDCP_ActTimeDL;

    var NR_RadioBearerId_Type v_RadioBearerId;

    var NR_SecurityActTime_Type v_SecurityActTime;

    var integer I;

    var NR_SecurityActTimeList_Type v_SecurityActTimeList;

    v_PdcpCountInfoList := f_NR_SS_PdcpCount_Get(p_CellId);

    for (I:=0; I < lengthof(v_PdcpCountInfoList); I:=I+1) {

      v_PdcpCountInfo := v_PdcpCountInfoList[I];

      if ((not ispresent(v_PdcpCountInfo.UL)) or (not ispresent(v_PdcpCountInfo.DL))) {

        FatalError(__FILE__, __LINE__, "missing PdcpCountInfo for UL or DL");

      }

      v_RadioBearerId := v_PdcpCountInfo.RadioBearerId;

      // Convert PdcpCount to integer value (UL an DL)

      v_PDCP_ActTimeUL.SQN := f_NR_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.UL);

      v_PDCP_ActTimeDL.SQN := f_NR_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.DL);

                // increment SQN for SRB1 (UL and DL)

                  if (ischosen(v_RadioBearerId.Srb) and (v_RadioBearerId.Srb == tsc_SRB1)) {

                    v_PDCP_ActTimeUL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeUL.SQN);

                    v_PDCP_ActTimeDL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeDL.SQN);

                  }

      // set activation time

      v_SecurityActTime.RadioBearerId := v_RadioBearerId;

      v_SecurityActTime.UL := v_PDCP_ActTimeUL;

      v_SecurityActTime.DL := v_PDCP_ActTimeDL;

      v_SecurityActTimeList[I] := v_SecurityActTime;

    }

    return v_SecurityActTimeList;

  }
After Change
function f_NR_RRC_CipherActTime_GetInitial(NR_CellId_Type p_CellId,

                                              boolean p_SQNIncrement := true) runs on NR_BASE_PTC return NR_SecurityActTimeList_Type

  {

    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList;

    var NR_PdcpCountInfo_Type v_PdcpCountInfo;

    var NR_PDCP_ActTime_Type v_PDCP_ActTimeUL;

    var NR_PDCP_ActTime_Type v_PDCP_ActTimeDL;

    var NR_RadioBearerId_Type v_RadioBearerId;

    var NR_SecurityActTime_Type v_SecurityActTime;

    var integer I;

    var NR_SecurityActTimeList_Type v_SecurityActTimeList;

    v_PdcpCountInfoList := f_NR_SS_PdcpCount_Get(p_CellId);

    for (I:=0; I < lengthof(v_PdcpCountInfoList); I:=I+1) {

      v_PdcpCountInfo := v_PdcpCountInfoList[I];

      if ((not ispresent(v_PdcpCountInfo.UL)) or (not ispresent(v_PdcpCountInfo.DL))) {

        FatalError(__FILE__, __LINE__, "missing PdcpCountInfo for UL or DL");

      }

      v_RadioBearerId := v_PdcpCountInfo.RadioBearerId;

      // Convert PdcpCount to integer value (UL an DL)

      v_PDCP_ActTimeUL.SQN := f_NR_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.UL);

      v_PDCP_ActTimeDL.SQN := f_NR_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.DL);

      if (p_SQNIncrement) {
          // increment SQN for SRB1 (UL and DL)

                  if (ischosen(v_RadioBearerId.Srb) and (v_RadioBearerId.Srb == tsc_SRB1)) {

                    v_PDCP_ActTimeUL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeUL.SQN);

                    v_PDCP_ActTimeDL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeDL.SQN);

                  }

      }
      // set activation time

      v_SecurityActTime.RadioBearerId := v_RadioBearerId;

      v_SecurityActTime.UL := v_PDCP_ActTimeUL;

      v_SecurityActTime.DL := v_PDCP_ActTimeDL;

      v_SecurityActTimeList[I] := v_SecurityActTime;

    }

    return v_SecurityActTimeList;

  }
5. Execution Log Files

5.1 HiSilicon Balong 5000
The HiSilicon Balong 5000 UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on NR band n41. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_1_1_1_1_LOG.html
PIXCIT settings used:

      TC_8_1_1_1_1_PIXIT.xml

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

References

	[1]
	R5s190215:   Supporting information for agreement of NR5GC testcase 8.1.1.1.1
                          This archive comprises:

                        -  html and text format execution log files for Keysight test platform



